سوال 11:

USE UniversityDB;
GO

SELECT 
    I.FirstName + ' ' + I.LastName AS [نام مدرس],
    COUNT(DISTINCT C.CourseID) AS [تعداد دوره‌ها],
    COUNT(DISTINCT E.StudentID) AS [تعداد کل دانشجویان],
    AVG(ISNULL(C.Price, 0)) AS [میانگین قیمت دوره‌ها],
    CASE 
        WHEN COUNT(DISTINCT C.CourseID) > 2 THEN N'Senior'
        ELSE N'Junior'
    END AS [سطح مدرس]
FROM 
    Instructors I
    JOIN Courses C ON I.InstructorID = C.InstructorID
    LEFT JOIN Enrollments E ON C.CourseID = E.CourseID
GROUP BY 
    I.InstructorID, 
    I.FirstName, 
    I.LastName
HAVING 
    COUNT(DISTINCT E.StudentID) >= 3;


سوال 12:

USE UniversityDB;
GO

SELECT 
    S.FirstName + ' ' + S.LastName AS [نام دانشجو],
    COUNT(LV.ViewID) AS [تعداد کل ویدیوهای دیده],
    COUNT(DISTINCT E.CourseID) AS [تعداد دوره‌های ثبت نام شده],
    CASE 
        WHEN COUNT(LV.ViewID) > 3 THEN 'Active'
        ELSE 'Inactive'
    END AS [وضعیت فعالیت]
FROM 
    Students S
    INNER JOIN Enrollments E ON S.StudentID = E.StudentID
    LEFT JOIN LessonViews LV ON S.StudentID = LV.StudentID
GROUP BY 
    S.StudentID, 
    S.FirstName, 
    S.LastName;


سوال 13:

نسخه صحیح کوئری:

SELECT
    c.CourseTitle,
    COUNT(e.StudentID) AS StudentCount
FROM Courses c
LEFT JOIN Enrollments e
    ON c.CourseID = e.CourseID
    AND e.EnrollDate >= '2024-01-01'  -- شرط تاریخ به اینجا منتقل شد
GROUP BY c.CourseTitle;


سوال 14:

کد Stored Procdure:

CREATE PROCEDURE sp_StudentsAboveAveragePayment
AS
BEGIN
    SET NOCOUNT ON;

    SELECT 
        S.StudentID,
        S.FirstName + ' ' + S.LastName AS FullName,
        SUM(P.Amount) AS TotalPayment
    FROM 
        Students S
        INNER JOIN Payments P ON S.StudentID = P.StudentID
    GROUP BY 
        S.StudentID, S.FirstName, S.LastName
    HAVING 
        SUM(P.Amount) > (
            -- محاسبه میانگین مجموع پرداخت‌های تمام دانشجویان
            SELECT AVG(SumPerStudent) 
            FROM (
                SELECT SUM(Amount) AS SumPerStudent 
                FROM Payments 
                GROUP BY StudentID
            ) AS StudentTotals
        );
END;
GO


سوال 15:

Stored Procedure:

CREATE PROCEDURE sp_ReportCourseRevenue
    @StartDate DATE,
    @EndDate DATE
AS
BEGIN
    SET NOCOUNT ON;

    SELECT 
        c.CourseTitle AS [عنوان دوره],
        COUNT(e.EnrollmentID) AS [تعداد ثبت نام‌ها],
        SUM(ISNULL(p.Amount, 0)) AS [مجموع پرداخت],
        CASE 
            WHEN SUM(ISNULL(p.Amount, 0)) >= 5000000 THEN 'High Revenue'
            ELSE 'Low Revenue'
        END AS [وضعیت دوره]
    FROM 
        Courses c
        INNER JOIN Enrollments e ON c.CourseID = e.CourseID
        LEFT JOIN Payments p ON e.StudentID = p.StudentID
    WHERE 
        e.EnrollDate BETWEEN @StartDate AND @EndDate
    GROUP BY 
        c.CourseID, c.CourseTitle;
END;
GO

سوال 16:

CREATE TRIGGER trg_HighValueStudent
ON Payments
AFTER INSERT, UPDATE
AS
BEGIN
    SET NOCOUNT ON;

    DECLARE @StudentID INT;
    DECLARE @TotalAmount INT;

    -- تعریف یک کرسر (Cursor) برای بررسی دانشجو هایی  که در این عملیات تغییر کرده‌اند
    DECLARE cur CURSOR FOR
        SELECT DISTINCT StudentID FROM inserted;

    OPEN cur;
    FETCH NEXT FROM cur INTO @StudentID;

    WHILE @@FETCH_STATUS = 0
    BEGIN
        -- محاسبه مجموع کل پرداخت‌های این دانشجو در جدول اصلی
        SELECT @TotalAmount = SUM(ISNULL(Amount, 0))
        FROM Payments
        WHERE StudentID = @StudentID;

        -- بررسی شرط و نمایش پیام
        IF @TotalAmount > 10000000
        BEGIN
            PRINT 'High Value Student';
            -- اگه میخوای بدونی کدوم دانشجو، خط زیر فعال شود
            -- PRINT 'High Value Student (Student ID: ' + CAST(@StudentID AS VARCHAR(10)) + ')';
        END

        FETCH NEXT FROM cur INTO @StudentID;
    END

    CLOSE cur;
    DEALLOCATE cur;
END;
GO